1. IDENTIFICATION

1.1 Digital-8-35-S-A

1.2 680 5-Bit Character Assembly Subroutines

1.3 November 17, 1965



2. ABSTRACT

These subroutines concentrate Teletype data by assembling serial-bit data into 5-bit characters and presenting the user with data similar to that obtained by using a 630 and scanner. They also add start and stop bits to 5-bit characters and transmit them in serial-bit fashion. Full duplex lines are assumed, but the subroutines can operate with half duplex if the user handles the expected echo.

3. REQUIREMENTS

3.1 Storage

The subroutines as presently coded occupy 400 octal locations plus space for internal buffering of the input and output characters and for the TTI instructions. In addition, space is used in memory page 0 and a limited number of autoindex registers are used as explained below. Within the limits described, the program can be placed anywhere in the first 4K of PDP-8 memory. The total amount of memory used, including the autoindex registers and the locations in page 0, is as follows:

$435_8 + 7n$

where n is the number of teletype lines to the next even multiple of eight lines if the number of lines is not already an even multiple of eight.

3.2 Subprograms and/or Subroutines

Digital-8-35-S-B
680 8-Bit Character Assembly Subroutines for reference or when the user's requirements include a mixture of 5-bit and 8-bit lines.

3.3 Equipment

Minimum configuration PDP-8
680 Data Communication System hardware
1 to 128 5-bit Teletype lines

3.4 Miscellaneous

3.4.1 The tag TT5BGN must be defined as the address of the start of the Teletype subroutines. It can be defined as anywhere in memory, but must be equivalent to the start of a PDP-8 memory page.

3.4.2 Four autoindex registers called T5AX1, T5AX2, T5AX3, and T5AX4 must be defined.

3.4.3 The tag TT5PG0 must be defined as the start of an area in memory page 0 where the necessary Teletype constants can be stored. An area of 26_8 registers must be reserved.

3.4.4 The tag T5OBF must be defined as the start of the area reserved for outputting the Teletype characters. It must be equal in length to the number of lines (even multiple of 8) attached to the particular set of subroutines. It can be anywhere in memory and need not start at the beginning of a memory page.

3.4.5 The tag T5OBF2 must be defined as an area equal in length to T5OBF. It is used for double-buffering the output characters to allow maximum output rate.

3.4.6 The tag T5IBF must be defined as the area for storing incoming Teletype characters and line numbers. It must be equal in length to twice the number of lines attached to the particular set of subroutines.

3.4.7 The tag T5IN must be defined as the start of the area used by the subroutines for generating the appropriate number of TTI instructions. It must be equal in length to three times the number of lines plus one register. Here again it need not be defined as the start of a memory page.

3.4.8 The tag TTCHAR must be defined as a single register in page 0.

3.4.9 In the interrupt service routine the following set or sets of instructions must appear:

```
T5SKP          /SKIP ON CLOCK FLAG
SKP            /TEST FOR NEXT INTERRUPT CAUSE
JMP T5DIS      /JUMP TO APPROPRIATE CLOCK INTERRUPT ROUTINE
```

Because of the speed necessary for Teletype handling, the checks for clock interrupts should be the first ones in the interrupt service interrogation loop; the link bit and accumulator contents should not be saved prior to interrogation of the appropriate clock flag. If necessary for other interrupts, the link and accumulator contents should be saved only after all clock interrupts have been checked.

3.4.10 Clock IOT's

The IOT's to test the clock for a 1 state, turn the clock on, and turn the clock off must be given the correct octal definitions:

Mnemonic	Clock 1	Clock 2	Clock 3	Clock 4
T5SKP	6421	6431	6441	6451
TT5ON	6424	6434	6444	6454
TT5OFF	6422	6432	6442	6452

4. USAGE

4.2 Calling Sequence

The pseudo command T5INIT must be executed before the instruction TT5ON and also before either of the other pseudo commands T5SOF or T5SIR is executed (See Sections 4.4.1, 4.4.2, and 4.4.3 for definitions of the pseudo commands.)

4.3 Switch Settings

None

4.4 Start up and/or Entry

Three pseudo commands for using this set of subroutines are provided to the main program. They are defined as jumps to subroutines and their definitions and instructions are included in the package. These are the only commands necessary in the main program for gathering and outputting the Teletype characters. The user should note that no subroutines are included for packing or unpacking the characters by word or even line number.

4.4.1 Teletype Initialize (T5INIT)

This command (which must be used only once in the main program) assumes that the user enters with the number of lines in the accumulator and that the register following the initialize command

contains the first line number for this type of Teletype line. This subroutine initializes all of the buffer areas, counters, and pointers, and generates the proper number of TTI instructions.

4.4.2 Skip if Output Free (T5SOF)

This instruction skips the next register in memory and transmits the character contained in register TTCHAR if the indicated output line is free. If the output line is not free, the instruction does not skip. The instruction requires that the line number over which the character is to be transmitted be in the accumulator at the time the instruction is issued. The pseudo command takes 24 μsec minimum time, and 42 μsec maximum time. The accumulator is cleared when exiting from the command.

4.4.3 Skip if Input Ready (T5SIR)

This instruction skips the next location in memory and returns with the line number in the accumulator and the character placed at TTCHAR if an input character is available. If no character is available, the instruction does not skip and the contents of the accumulator equal −1. Only the low order 5 bits of the character at TTCHAR should be used, as additional bits representing the stop codes are also present in the character. The user should note that the bit structure of the character is reversed from DEC's standard Teletype code. (For example, the character 0 does not appear as 15 in the low order 5 bits, but as 26. This special consideration may be important if the user is setting up any necessary conversion tables.)

If no character is available, 15 μsec are used by the pseudo instructions; if a character is available, 37.5 μsec are used; and if the end of the storage area is reached, a maximum of 48 μsec are used by the instruction.

5. RESTRICTIONS

5.1 Status Active Registers

The autoindex registers defined as T5AX1, T5AX2, T5AX3, and T5AX4 must not be disturbed after the pseudo operation T5INIT.

6. DESCRIPTION

6.1 Discussion

These subroutines are designed to accumulate 5-bit Teletype characters to and from multiple Teletype lines connected to a PDP-8. They handle input data in serial-bit format and present the user with character and line identification. The user presents the routines with line identification and character format data and they transmit the information in serial-bit format.

Most of the PDP-8 memory is available for data buffering and for packing. A large proportion of the time, however, is used mainly in buffering the Teletype lines themselves. Assuming only minor data handling is necessary before transmission (possibly to a larger computer), present estimates indicate that the user could handle 128 5-bit lines at 50 baud. Exact timing information is shown in Section 9. The user should note that the programming described involves the handling of the Teletype lines only and does not include any packing or unpacking of words, lines, or messages. The main program communicates with the Teletype subroutines via a group of pseudo commands which are described fully in Section 4.4 with examples of their usage in Section 6.2.

If the user's requirements include mixed speeds of 5-bit lines, these subroutines must be duplicated for each line speed. Or, if a mixture of 5-bit and 8-bit lines is required, it is necessary that the 8-Bit Character Assembly Subroutines (Digital-8-35-S-B) be included with the user's programs and the 5-Bit Character Assembly Subroutines.

6.2 Examples and/or Applications

6.2.1 To initialize the subroutines, coding similar to the following should appear in the user's program:

```
TAD     NUMLIN          /GET NUMBER OF LINES.
T5INIT                  /INITIALIZE SUBROUTINES.
SLN                     /STARTING LINE NUMBER.
ION                     /ENABLE INTERRUPTS.
TT5ON                   /TURN ON CLOCK.
```

6.2.2 To output a character, coding similar to the following should appear:

```
TAD     CHARAC          /GET OUTPUT CHARACTER.
DCA     TTCHAR          /FOR OUTPUT SUBROUTINE.
TAD     LINE NO         /GET LINE NUMBER.
T5SOF                   /OUTPUT, SKIP IF FREE.

JMP     OUTNA           /OUTPUT NOT FREE.
CONTINUE                /CHARACTER ACCEPTED, CONTINUE.
```

6.2.3 To test for an input character available, coding similar to the following should appear:

```
T5SIR                   /CHECK FOR INPUT.
JMP     .-1             /WAIT FOR A CHARACTER.
DCA     SAVLIN          /SAVE LINE NUMBER.
TAD     TTCHAR          /GET CHARACTER INPUT.
AND     THREE7          /37, CLEAR STOP BIT.
```

7. METHODS

7.1 Discussion

7.1.1 Input Character Assembly

The 5-Bit Character Assembly Interrupt Subroutine executes a TTI instruction for each line selected every clock interrupt. The program then scans one fourth of the TTI character assembly words for fully assembled input characters. When an assembled input character is found, the program shifts off the start bit, stores the character and line number in the input buffer, zeros the TTI status word and resets the character assembly word to 0400. Note that bit 3 is initially set to a 1 and the rest of the character assembly word is zero. As the input character is assembled bit by bit, the character assembly word is shifted one position to the right for the start bit, each data bit, and the stop bit. When the bit that was initially in the character assembly word can be set into the link by a RTR, the character is fully assembled.

7.1.2 Output Character Handling

Initially, the pseudo operation T5SOF adds start and stop bits to the output characters and places the characters in the second output buffer (T5OBF2). Eventually, the interrupt subroutine transfers the characters from the second output buffer to the first output buffer (T5OBF). In the interrupt subroutine, the program outputs on one eighth of the lines selected every clock interrupt. That is, for any one line the program outputs a data bit every eight clock interrupts. If the first output buffer location for a line is zero, there is no output on that line. After 56 clock interrupts (7 bit times), the program halts the output process and utilizes each of the next four interrupts (one half bit time) to scan one fourth of the second

output buffer for new output characters. Again, if the second output buffer location for a line is zero, there is no output. When a location is found that is non-zero, the character is placed in the first output buffer and the second output buffer location is set to zero.

8. FORMAT

8.1 Input Data (T5SIR)

If the pseudo operation T5SIR skips, the input data is the following format:

8.1.1 The accumulator contains the line number.

8.1.2 The lower five bits of the register TTCHAR contain the input character. (See NOTE in Section 8.3)

8.3 Output Data (T5SOF)

The user presents the pseudo operation T5SOF with output characters in the following format:

8.3.1 The lower five bits of the register TTCHAR contain the output character.

8.3.2 The accumulator contains the number of the line on which the character is to be output.

NOTE: As mentioned in Section 4.4.3, the bit structure of the 5-bit codes is reversed from standard. These subroutines present the user with this reversed code and similarly expect the user to present them with the reversed code.

9. EXECUTION TIME

9.1 Minimum

9.2 Maximum

9.3 Average

The table below indicates the percentages of machine time used for two speeds of 5-bit systems and is as accurate as presently possible. Any additional features which may be required for the Teletype handling would add appreciably to the times shown:

TIMING TABLE

Numbers indicate the percentage of available machine time used in the average case.

No. of Lines	5-Bit 50 Baud	5-Bit 75 Baud
32	20.0%	30.0%
64	35.1%	52.7%
96	50.3%	75.5%
128	65.5%	98.3%

9.4 Timing Equations

9.4.1 50 Baud Rate

Where n = the number of lines, the 5-bit subroutines require an average time of 11.85n + 120 μsec. Clock flags (at 50 baud) occur every 2500 μsec.

9.4.2 75 Baud Rate

The percentages for 75 baud are merely 1.5 x 50 baud rate. Clock flags occur every 1667 μsec.

10. PROGRAM

10.3 List of Items and Pseudo Commands

10.3.1 List of Items

TT5BGN	Start of subroutine, must be equated to the start of a page. Area includes 2 pages.
T5AX1	Autoindex register.
T5AX2	Autoindex register.
T5AX3	Autoindex register.
T5AX4	Autoindex register.
TT5PG0	Start of constant area in page 0.
T5OBF	Start of output buffer (Length = n).
T5OBF	Start of second output buffer (Length = n).
T5IBF	Start of input buffer (Length = 2n).
T5IN	Start of TTI area (Length = 3n + 11).
TTCHAR	Character area page 0 (Single register).

10.3.2 List of Pseudo Operations

Operation	Meaning	Times (User's) Min.	Av.	Max.
T5INIT	Initialize	N.A.		
T5SOF	Skip if output free	24	-	42
T5SIR	Skip if input ready	15	37.5	48

10.4 Program Listing

```
                /TYPE 680 TELETYPE LINE MULTIPLEXER
                /CHARACTER ASSEMBLY ROUTINE
                /LMH  910/15/65  5 BIT

                TTI=6402                /TELETYPE INPUT COMMAND
                TTO=6404                /TELETYPE OUTPUT COMMAND
                TTCL=6411               /CLEAR LINE REGISTER
                TTRL=6414               /READ LINE REGISTER
                TTSL=6412               /SET LINE REGISTER, CLR AC
                TT5ON=6424              /TURN CLOCK ON
                TT5OFF=6422             /TURN CLOCK OFF
                T5SKP=6421              /SKIP ON CLOCK FLAG
                TTINCR=6401             /INCREMENT LINE REGISTER

                /680 LINE MULTIPLEXER
                /LIST OF ITEMS
                T5IBF=7200
                T5OBF2=7000
                T5OBF=6600
                T5IN=5600
                TT5PG0=145
                T5AX1=10
                T5AX2=11
                T5AX3=12
                T5AX4=13
                TT5BGN=5200
                TTCHAR=177

                *TT5PG0
0145  0000  T5INFL,   0             /INPUT READY FLAG
0146  7177  T5BFK,    T5IBF-1       /TO RESET INPUT BUFFER POINTER
0147  0000  T5NL,     0             /-NUMBER OF LINES
0150  5400  T5SOUT,   T5OUTS        /SKIP IF OUTPUT FREE
0151  5423  T5SIN,    T5INS         /SKIP IF INPUT READY
0152  5447  T5GO,     T5GOS         /INITIALIZE ROUTINE
0153  6600  T5OUTK,   T5OBF         /POINTER TO 1ST OUTPUT BUFFER
0154  7774  T5CNT1,   -4            /HOLDS MAJOR LOOP COUNTER
0155  0000  T5CNT2,   0             /MINOR LOOP COUNTER
0156  0000  T5CNT3,   0             /COUNTER FOR INPUT BUFFER
0157  0177  T5K10,    177           /FOR ANDING
0160  7000  T5K36,    T5OBF2        /2ND OUTPUT BUFFER
0161  0000  T5CNT5,   0             /OUTPUT COUNTER
0162  0000  T5CNT6,   0             /7 BIT COUNTER
0163  7770  T5K2,   -10         /TO RESET BIT COUNTER
0164  5600  T5K3,   T5IN                 /RESET INPUT TTI POINTER
0165  7776  T5K5,     -2            /FOR SUBTRACTION
0166  0400  T5K6,   400         /TO RESET 5 BIT ASSEMBLY WORD
0167  6600  T5K7,     T5OBF         /K FOR 1ST OUTPUT BUFFER
0170  5221  T5K8,     T5COM         /TO ENTER COMMON ROUTINE
0171  0000  T5K9,     0             /LINE NUMBER -1
0172  6577  T5K9A,    T5OBF-1       /FOR CLEARING
0173  5267  T5K9B,    T5CM1A        /TO AVOID OUTPUTTING
0174  5361  T5K9C,    JMP T5CM10 /TO SET OUTPUT BUFFER FROM DOUBLE BUFFER
0175  5221  T5K9D,    T5COM         /FOR NORMAL RETURN
0176  7000  T5K9E,    NOP           /TO DO INPUT ONLY
```

```
                *TT5BGN
                /MULTIPLE LEVEL INTERRUPT ROUTINE
                /ALLOWS MULTIPLE LEVEL INTERRUPT TO THIS ROUTINE
                /AND UNLIMITED

5200  2366  T5DIS, ISZ T5LC          / LEVEL COUNTER
5201  5216        JMP T5DIS3          /2ND LEVEL INTERRUPT
5202  3367        DCA T5SA            / SAVE ACCUMULATOR
5203  7010        RAR       /GET LINK
5204  3370        DCA T5SVLK          /SAVE LINK
5205  1000        TAD Z 0             / INTERRUPT ADDRESS
5206  3371        DCA T5SV0           / SAVE ADDRESS
5207  6414           TTRL                  /READ LINE NUMBER
5210  3372           DCA T5SVLN            /SAVE LINE NUMBER
5211  6424           TI5ON                 /TO CLEAR CLOCK FLAG ONLY
5212  6001  T5DIS2,  ION                   /RE-ENABLE PROGRAM INTERRUPT
5213  1171           TAD T5K9              /STARTING LINE-1
5214  6413           TTSL+1                /SET LINE REGISTER, CLR AC
5215  5564        JMP I Z T5K3         /SET LINE REGISTER, CLR AC

                /2ND LEVEL INTERRUPT
5216  6424  T5DIS3,  TI5ON                 /CLEAR CLOCK FLAG
5217  6001           ION                   /RE-ENABLE PROGRAM INTERRUPT
5220  5400           JMP I Z 0             /RETURN TO THE MAIN PROGRAM

                /RETURN FROM INPUT TTI LOOP
5221  1373  T5COM,    TAD T5MNC           /-NO. OF LINES/8
5222  3155             DCA Z T5CNT2        /MINOR LOOP COUNTER
5223  1375             TAD T5LN            /LINE NUMBER
5224  6413          TTSL+1              / SET LINE NUMBER
5225  1553  T5COM0,    TAD I Z T5OUTK      /OUTPUT WORD
5226  7450             SNA                 /CHARACTER AVAILABLE
5227  5351             JMP T5COM3          /NOTHING TO TRANSMIT
5230  6405             TTO+1               /INCREMENT AND TRANSMIT
5231  3553             DCA I Z T5OUTK      /RESTORE CHARACTER
5232  2153  T5COM1,    ISZ Z T5OUTK        /UPDATE OUTPUT POINTER
5233  2155             ISZ Z T5CNT2        /ARE ONE-EIGHTH OF LINES DON
5234  5225             JMP T5COM0          /CHECK NEXT OUTPUT LINE
5235  6414             TTRL                /READ LINE NUMBER
5236  3375             DCA T5LN            /SAVE LINE NUMBER
5237  1374  T5CM1A,    TAD T5MNC2          /NO OF LINES/4
5240  3155             DCA T5CNT2          /MINOR LOOP COUNTER
5241  2010  T5COM2,        I SZ T5AX1          / ADVANCE FOR NEXT INPUT
5242  1410           TAD I Z T5AX1       /CHARACTER ASSEMBLY WORD
5243  7112           CLL RTR             / PUT BIT 10 IN LINK
5244  7450              SZL                   /CHARACTER NOT COMPLETED
5245  5326              JMP T5COM6            /STORE CHARACTER
5246  7200              CLA                   /CLEAR AC FOR TAD
5247  7000  T5COM3,     NOP                   /OR JMP T5CM10
5250  2010              ISZ Z T5AX1           /UPDATE FOR NEXT INPUT LINE
5251  2376           ISZ T5LN2           / UPDATE LINE NUMBER
5252  2155              ISZ T5CNT2            /ARE ONE-FOURTH OF LINES
5253  5241              JMP T5COM2            /CHECK NEXT LINE
5254  2154  T5COM4,     ISZ T5CNT1            /HAVE ALL INPUT LINES BEEN
5255  5310              JMP T5COM5            /RESET AND DISMISS
5256  1154           TAD Z T5K3          /T5IN
```

```
5257  3010            DCA Z T5AX1             /RESET TTI POINTER
5260  1171          TAD Z T5K9          /START LINE-1
5261  7001          IAC       /SET TO FIRST LINE
5262  3376            DCA T5LN2               /RESET LINE NUMBER
5263  1377            TAD T5K2A               /-4
5264  3154            DCA T5CNT1              /INPUT CHECK COUNTER
5265  2161            ISZ Z T5CNT5            /HAVE ALL OUTPUT LINES BEEN
5266  5310            JMP T5COM5              /RESET AND DISMISS
5267  1165            TAD Z T5K5              /-2
5270  3161            DCA Z T5CNT5            /RESET COUNTER
5271  1171            TAD Z T5K9              /START LINE-1
5272  3375            DCA T5LN                /RESET LINE NUMBER
5273  2162            ISZ Z T5CNT6            /ENDING 7TH BIT?
5274  5353            JMP T5COM9              /NO RESET NORMALLY
5275  1163            TAD T5K2                /-10
5276  3162            DCA Z T5CNT6            /RESET COUNTER
5277  2161            ISZ Z T5CNT5            /ADD 1 TO COUNTER
5300  1172            TAD Z T5K9A             /T50BF-1
5301  3013            DCA Z T5AX4             /OUTPUT POINTER
5302  1160            TAD Z T5K36             /T50BF2
5303  3153            DCA T50UTK              /2ND BUFFER POINTER
5304  1173            TAD Z T5K9B             /SPECIAL ADDRESS, T5CM1A
5305  3170            DCA Z T5K8              /RESET ADDRESS
5306  1174            TAD Z T5K9C             /JMP T5CM1.0
5307  3247            DCA T5COM3              /SET TO DO OUTPUT
5310  6002  T5COM5,   IOF                     /TURN OFF INTERRUPT
5311  7240            STA                     /-1
5312  1366            TAD T5LC                /LEVEL COUNTER
5313  3366            DCA T5LC                /RESTORE LEVEL COUNTER
5314  1366            TAD T5LC                /LEVEL COUNTER
5315  7700            SMA CLA                     /RESTORE AC, ETC.
5316  5212            JMP T5DIS2             /CHECK INPUT AGAIN, ETC.
5317  1372            TAD T5SVLN              /LINE NUMBER
5320  6413            TTSL+1                    /SET LINE REGISTER, CLR AC
5321  1370            TAD T5SVLK             /PICK UP LINK
5322  7104            CLL RAL                 /RESTORE LINK
5323  1367            TAD T5SA                /RESTORE AC
5324  6001            ION                     /RE-ENABLE PROGRAM INTERRUPT
5325  5771            JMP I T5SV0             /RETURN TO THE MAIN PROGRAM
5326  7112  T5COM6,        C LL RTR              / REMOVE START CODE
5327  7012          RTR
5330  3411            DCA I Z T5AX2           /STORE CHARACTER
5331  1376            TAD T5LN2               /LINE NUMBER
5332  3411            DCA I Z T5AX2           /STORE LINE NUMBER
5333  1010            TAD Z T5AX1             /TTI POINTER
5334  1165            TAD Z T5K5              /-2
5335  3010            DCA Z T5AX1             /RESET POINTER
5336  3410            DCA I Z T5AX1           /ZERO STATUS AND COUNTER
5337  1166            TAD Z T5K6              /WORD TO RESTORE ASSEMBLY WD
5340  3410            DCA I Z T5AX1           /RESET CHARACTER ASSEMBLY WD
5341  2145            ISZ Z T5INFL            /SET INPUT READY FLAG
5342  2156            ISZ Z T5CNT3            /HAS END OF BUFFER BEEN REAC
5343  5247            JMP T5COM3              /CONTINUE
5344  1146  T5COM7,   TAD Z T5BFK             /T5IBF-1
5345  3011            DCA Z T5AX2             /RESET INPUT BUFFER ADDRESS
5346  1147            TAD T5NL                /-NUMBER OF LINES
5347  3156            DCA Z T5CNT3            /RESET LENGTH COUNTER
5350  5247            JMP T5COM3              /CONTINUE
```

```
5351  6401  T5COM3,   ITINCR                 /INCREMENT LINE NUMBER
5352  5232             JMP T5COM1             /CONTINUE
5353  1167  T5COM9,   TAD Z T5K7             /T50BF
5354  3153             DCA Z T50UTK           /RESET OUTPUT POINTER
5355  1175             TAD Z T5K9D            /NORMAL ADDRESS, T5COM
5356  3170             DCA Z T5K8             /RESET ADDRESS
5357  1176             TAD Z T5K9E            /NOP
5360  5307             JMP T5COM5-1           /CONTINUE
5361  1553  T5CM10,        T AD I Z T50UTK        /2ND BUFFER CHARACTER
5362  3413             DCA I Z T5AX4          /STORE IN 1ST BUFFER
5363  3553             DCA I T50UTK           /CLEAR 2ND BUFFER
5364  2153             ISZ T50UTK             /UPDATE POINTER
5365  5250             JMP T5COM3+1           /CONTINUE

                /CONSTANTS
5366  7777  T5LC,     -1                     /INTERRUPT LEVEL COUNTER
5367  0000  T5SA,     0                      /SAVE ACCUMULATOR
5370  0000  T5SVLK,   0                      /SAVE LINK
5371  0000  T5SV0,    0                      /SAVE PROGRAM COUNTER
5372  0000  T5SVLN,   0                      /SAVE LINE NUMBER
5373  0000  T5MNC,    0                      /-NO OF LINES/8
5374  0000  T5MNC2,   0                      /-NO OF LINES/4
5375  0000  T5LN,     0                      /LINE NUMBER FOR OUTPUT
5376  0000  T5LN2,    0                      /LINE NUMBER FOR INPUT
5377  7774  T5K2A,    -4                     /TO RESET MAJOR LOOP COUNTER
                /PSEUDO-OPERATIONS
                *IT5BGN+200
                /SKIP IF OUTPUT IS FREE AND TRANSMIT CHARACTER AT ITCHAR,
                /DON'T SKIP LINE NUMBER MUST BE IN AC. 24US MIN., 42US MAX.
                T5SOF=JMS I Z T5SOUT
5400  0000  T50UTS,   0
5401  0157             AND Z T5K10            /177
5402  1217             TAD T5SL               /-STARTING LINE NUMBER
5403  1160             TAD Z T5K36            /OUTPUT BUFFER ADDRESS
5404  3220             DCA T5WA               /WORK AREA
5405  1620             TAD I T5WA             /OUTPUT CHARACTER
5406  7640             SZA CLA                /SKIP IF FREE
5407  5600             JMP I T50UTS           /EXIT
5410  1177             TAD Z ITCHAR           /PICK UP CHARACTER
5411  0221             AND T5K11              /5 BITS ONLY
5412  1222             TAD T5K12              /140 FOR STOP CODE
5413  7104             CLL RAL                /CREATE START CODE
5414  3620             DCA I T5WA             /STORE CHARACTER IN TABLE
5415  2200             ISZ T50UTS             /INDEX EXIT
5416  5600             JMP I T50UTS           /EXIT
5417  0000  T5SL,     0                      /-STARTING LINE NUMBER
5420  0000  T5WA,     0                      /WORK AREA
5421  0037  T5K11,    37                     /FOR 5 BIT CODE
5422  0040  T5K12,    40                     /FOR STOP CODE

          /SKIP IF CHARACTER AVAILABLE AND RETURN WITH LINE NUMBER IN&
          /CHAR AT ITCHAR.  OTHERWISE DO NOT SKIP
          /15US MIN., 48US MAX., 37.5US NORMAL IF READY
          T5SIR=JMS I Z T5SIN
5423  0000  T5INS,    0
5424  6002            IOF
5425  7240            CLA CMA              / SET AC FOR TAD-1
```

```
5426  1145          TAD Z T5INFL        /INPUT FLAG COUNTER-1
5427  7510          SPA       /SOMETHING AVAILABLE
5430  5244          JMP T5INON          /EXIT
5431  3145             DCA Z T5INFL          /RESTORE FLAG COUNTER
5432  2246             ISZ T5CNT4            /END OF BUFFER?  START AT -#
5433  5240             JMP .+5               /GET CHARACTER
5434  1147             TAD Z T5NL            /-NUMBER OF LINES
5435  3246             DCA T5CNT4            /RESET COUNTER
5436  1146             TAD Z T5BFK           /BUFFER ADDRESS-1
5437  3012             DCA Z T5AX3           /RESET ADDRESS
5440  1412             TAD I Z T5AX3         /PICK UP CHARACTER
5441  3177             DCA Z TICHAR          /STORE CHARACTER
5442  1412             TAD I Z T5AX3         /PICK UP LINE NO.
5443  2223             ISZ T5INS             /INDEX EXIT
5444  6001  T5INON,        I ON
5445  5623             JMP I T5INS           /EXIT
5446  0000  T5CNT4,   0                      /-NUMBER OF LINES
            /INITIALIZATION ROUTINE
            /ENTER WITH NUMBER OF LINES IN AC
            /FORMAT   T5INIT
            /         1ST LINE NO.

            T5INIT=JMS I Z T5GO
5447  0000  T5GOS,    0
5450  0355             AND T5K14             /377
5451  3147             DCA Z T5NL            /NO. OF LINES
5452  1147             TAD Z T5NL            /NO. OF LINES
5453  0356             AND T5K15             /7
5454  7640             SZA CLA               /MULTIPLE OF 8
5455  1357             TAD T5K16             /10
5456  1147             TAD Z T5NL            /NO. OF LINES
5457  0360             AND T5K17             /370
5460  7041             CIA                   /TWO'S COMP. NUMBER OF LINES
5461  3147             DCA Z T5NL            /-N, CONSTANT
5462  1147  T5G01,     TAD Z T5NL            /-N
5463  3156          DCA Z T5CNT3        /COUNTER
5464  1361             TAD T5K20             /T5IN-1
5465  3010             DCA Z T5AX1           /TO STORE TTI TABLE
5466  1362             TAD T5K21             /T5OBF-1
5467  3011             DCA Z T5AX2           /TO CLEAR OUTPUT AREA
5470  1373             TAD T5K37             /T8OBF2-1
5471  3012             DCA T5AX3             /TO CLEAR DOUBLE BUFFER
5472  1147             TAD Z T5NL            /-N
5473  3246             DCA T5CNT4            /FOR COUNTING
5474  1363  T5G02,     TAD T5K22             /TTI+INCR
5475  3410             DCA I Z T5AX1         /STORE TTI
5476  3410             DCA I Z T5AX1         /CLEAR STATUS WORD
5477  1166          TAD Z T5K6          /ASSEMBLY RESET WORD
5500  3410             DCA I Z T5AX1         /RESET ASSEMBLY WORD
5501  3411             DCA I Z T5AX2         /ZERO OUTPUT WORD
5502  3412             DCA I Z T5AX3         /CLEAR DOUBLE BUFFER
5503  2246          ISZ T5CNT4          /COUNTER
5504  5274             JMP T5G02             /DO NEXT LINE
5505  1364             TAD T5K24             /JMP I Z T5K8
5506  3410             DCA I Z T5AX1         /STORE FINAL JUMP
5507  1147             TAD Z T5NL            /-N
5510  7012             RTR                   /-N/4
```

```
5511  7010          RAR                     /-N/8
5512  0365          AND I5K25               /17
5513  1366          TAD I5K26               /7760, MAKE NUMBER NEGATIVE
5514  3767          DCA I I5K27             /I5MNC
5515  1767          TAD I I5K27             /I5MNC
5516  1767          TAD I I5K27             /I5MNC -N/4
5517  3774          DCA I I5K38             /I5MNC2
5520  7240          STA                     /-1
5521  3246          DCA I5CNT4              /SET CNTR TO SKIP 1ST TIME
5522  1146          TAD Z I5BFK             /I5IBF-1
5523  3011        DCA Z I5AX2         /SET INPUT BUFFER POINTER
5524  1370          TAD I5K28               /-4
5525  3154          DCA Z I5CNT1            /MAJOR LOOP COUNTER
5526  1165          TAD Z I5K5              /-2
5527  3161          DCA Z I5CNT5            /OUTPUT COUNTER
5530  1164          TAD Z I5K3              /I5IN+1
5531  3010          DCA Z I5AX1             /SET ITI POINTER
5532  1167          TAD I5K7                /I5OBF
5533  3153          DCA Z I5OUTK            /SET OUTPUT BUFFER POINTER
5534  7240          STA                     /-1
5535  1647          TAD I I5GOS             /STARTING LINE NUMBER
5536  3171          DCA Z I5K9              /STARTING LINE NO-1
5537  1171          TAD Z I5K9              /STARTING LINE -1
5540  7040          CMA                     /MAKE NEGATIVE
5541  3217          DCA I5SL                /-STARTING LINE NUMBER
5542  3145          DCA Z I5INFL            /CLEAR INPUT FLAG COUNTER
5543  7240          STA                     /-1
5544  3771          DCA I I5K35             /I5LC, RESET INTERRUPT LEVEL
5545  2247          ISZ I5GOS               /INDEX EXIT
5546  1372          TAD I5K35A              /-7
5547  3162          DCA Z I5CNT6            /SET SPECIAL 5-BIT COUNTER
5550  1175          TAD Z I5K9D             /I5COM
5551  3170          DCA Z I5K8              /ITI RETURN
5552  1176          TAD Z I5K9E             /NOP
5553  3775          DCA I I5K40             /I5COM3
5554  5647          JMP I I5GOS             /EXIT

              /CONSTANTS
5555  0377  I5K14,     377                 /FOR ANDING
5556  0007  I5K15,     7                   /FOR EVEN MULTIPLE OF 8
5557  0010  I5K16,     10                  /FOR EVEN MULTIPLE OF 8
5560  0370  I5K17,     370                 /FOR EVEN MULTIPLE OF 8
5561  5577  I5K20,     I5IN-1              /FOR STORING ITI'S
5562  6577  I5K21,     I5OBF-1             /FOR OUTPUT AREA
5563  6403  I5K22,  ITI+1              / ITI
5564  5570  I5K24,     JMP I Z I5K8        /FOR FINAL JUMP
5565  0017  I5K25,     17                  /FOR -N/8
5566  7760  I5K26,     7760                /FOR MAKING NEGATIVE
5567  5373  I5K27,     I5MNC               /FOR -N/8
5570  7774  I5K28,     -4                  /FOR MAJOR LOOP COUNTER
5571  5366  I5K35,     I5LC                /FOR INTERRUPT LEVEL COUNTER
5572  7771  I5K35A,    -7                  /FOR 5-BIT COUNTER
5573  6777  I5K37,     I5OBF2-1            /FOR DOUBLE BUFFER
5574  5374  I5K38,     I5MNC2              /FOR -N/4
5575  5247  I5K40,     I5COM3              /FOR SWITCH
```

11 DIAGRAMS

11.1 Flow Charts




4
T5COM2
IS INPUT CHAR FULLY ASSEMBLED ?
YES
NO
CLA
T5COM3
NOP OR JMP T5CM10
8
T5CM10
PICK UP OUTPUT 2ND BUFFER PUT IT IN 1ST AND CLR 2ND BUFFER
INCREMENT 2ND BUFFER POINTER
ADVANCE INPUT POINTERS AND LINE NO.
TESTED 1/4 OF LINES FOR INPUT ?
NO
YES
T5COM4
HAVE ALL LINES BEEN CHECKED FOR INPUT ?
NO
7 T5COM5
YES
RESET TTI POINTER LINE NUMBER AND INPUT CHECK
HAVE ALL OUTPUT LINES BEEN CHECKED ?
NO
YES
RESET T5CNT5 TO -2
T5COM6
REMOVE START CODE
STORE CHARACTER AND LINE NUMBER
ZERO STATUS AND COUNTER LOAD SET CHAR ASSEM FLAG
SET INPUT READY FLAG
8 T5COM3
RESET BUFFER ADDRESS AND INDEX INPUT
YES
END OF INPUT BUFFER ?
NO
LD AC WITH A NOP
5
RESET T5K8 TO T5COM
T5COM9
RESET OUTPUT BUFFER POINTER
NO
ENDING 7TH BIT ?
YES
(T5CNT6)=-10
RESET T5CNT6 T5CNT + 1
LD AC WITH T5CM10
RESET T5K8 TO T5CM1A
AX4=OBF-1
T5OUTK=OBF2
RESET OUTPUT BUFFER POINTER



12. REFERENCES

12.1 Other Library Programs

Digital-8-35-S-A
680 5-Bit Character Assembly Subroutines